88 Optimization Tips

by Francesco Balena

You can’t reduce the art of optimization to a mere collection of tips. But you can speed apps by following these suggestions.

Microsoft’s release of a compiler for Visual Basic has raised expectations for faster execution of VB programs. Unfortunately, however efficient a compiler is, and however powerful the hardware is, some code always runs too slowly—at least according to your customer.

To get that code running as fast as possible, you can follow a few guidelines that help grease the skids. With only a few exceptions, these guidelines are valid whether you compile your application into native code or whether you still use an older version of the language that only supports p-code generation. 

When I wrote down all the tips and optimization techniques I knew of, I ended up with more than 100 items, 88 of which I’ve included in this article. (You can download the rest from the free, Registered Level of The Development Exchange. See the Code Online box at the end of the article for details.) This is by no means an exhaustive list, but it’s a good starting point. Use it as a sort of checklist. When you’re trying to speed up an inefficient application, check whether it follows these guidelines and focus your efforts on the portions of the code that don’t comply with these rules.

All these tips apply to VB3, VB4 16- and 32-bit, and VB5, unless noted differently. VB3 programmers don’t have to worry about objects, OLE servers, or RDO. Similarly, VB3 and VB4 users may jump over VB5-specific features, such as ODBCDirect and native code compilation settings. The companion code has been tested under VB5 only; you should revise API calls for VB3 and VB4/16. 

One final note: I’m not fond of benchmarks, and you won’t see many absolute timings in my articles. However, when discussing optimization techniques, I can’t help comparing the relative performances of similar techniques. I’ve run all relative benchmarks on VB5 native compiled code on a Pentium 120 MHz system under Windows 95. These relative timings might differ greatly under older versions of VB or VB5’s p-code and will most likely differ if performed on a different system or under a different version of Windows. 

Variables: Constantly use these rules

Always use Integer and Long rather than Single, Double, or Currency variables, if possible. Math operations on Integer and Long values use single CPU instructions, and are generally much faster. 

**

Use Single rather than Double variables if you’re satisfied with their lower precision; Single variables take less memory and sometimes work faster in math expressions.

**

Don’t use Variants unless you really need their unique capabilities, such as their ability to store Empty or Null values. Each Variant takes 16 bytes (as opposed to the eight bytes taken by a Double or a Currency value) and is generally slower (see Figure 1). 

Figure 1: Avoid Unnecessary Waste of Memory. Learn how many bytes each VB data type takes so you don’t use more memory than strictly needed. Conventional strings and variants waste bytes for auxiliary data. Fixed-length strings conserve memory but are slower than variable-length strings. Under VB5 and VB4/32, all strings store Unicode characters, which take two bytes each.
**

Add an Option Explicit directive to the beginning of each module, so you don’t run the risk of inadvertently using an undeclared Variant variable. Alternatively, use a DefLng A-Z directive so all undeclared variables are of type Long. Add a warning to the top of the module to ensure that code editors do not overlook this directive.

**

Referencing a static local variable in a procedure is two to three times slower than a regular local, dynamic variable. If you want to speed up your procedure, convert all static variables into module-level variables. The only drawback to this approach is that the procedure becomes less self-contained, and if you wish to reuse it in another project, you must remember to copy and paste the module-level variable.

**

Referencing a variable declared at module level is faster than a global variable declared in a separate BAS module. If you don’t need to share a variable among all the forms and modules of an application, declare it in the only form or module that uses it.

Explicit and implicit conversions

Beware of Int and Val functions because they always return a floating-point value. If you’re dealing with Integer or Long values, you should use CInt and CLng functions. You can choose among four approaches for assigning the contents of a text-box control to an Integer or Long variable:

result% = Val(Text1.Text)

result% = Int(Text1.Text)

result% = CInt(Text1.Text)

result% = Text1.Text

Each statement in this code is slightly faster than the previous one; the last one is about 30 percent faster than the first one.

**

Use “\” instead of “/” when performing divisions between Integer variables. The “/” operator returns a Single value, so this seemingly efficient line actually requires three implicit conversions—two for converting the operand from Integer to Single (to prepare the Integer variables for the division), and one to convert the result from Single to Integer (to complete the assignment):

C% = A% / B%

If you use the “\” operator, you don’t incur this overhead, and the division itself is faster. While we’re on this topic, keep in mind that the “/” operator returns a Double value if at least one of its operands is a Double. Therefore, if you want the highest precision when dividing two Integer or Single values, you should manually coerce one of them into Double type:

' this prints 0.3333333

Print 1 / 3

' this prints 0.333333333333333

Print 1 / 3#

**

VB stores symbolic and literal constants internally using the simplest data type. This means that most common numeric constants—such as zero or one—are stored as Integers. If you use these constants in floating-point expressions, you can speed up your code using a constant of appropriate type:

value# = value# + 1#

This forces the compiler to store the constant in Double format, saving one implicit conversion at run time. Alternatively, you can declare and use a symbolic constant of a well-defined type:

Const ONE As Double = 1

Math lessons

The fastest functions are those you can evaluate in advance. For instance, storing the factorials of all numbers in the range of one to 100 into an array once at the beginning of the program is much faster than evaluating a function thousands of times during the life of the application. You might encounter problems with this technique: you might end up evaluating more values than you require during execution, and your routine will not be self-contained anymore because it depends on a global array. This makes reusing your routine a bit more difficult. You can work around both of these problems using Static data within the routine (see Listing 1).

[Listing 1] VB3, VB4 16/32, VB5

Function Factorial(ByVal n As Long) As Double

   ' optimized and self-contained routine that evaluates 

   ' the factorial of a number in the range 0-170

   Dim i As Long

   Static fact(170) As Double

   Static maxValue As Long

   ' maxValue holds the highest value of "N" ever

   ' passed to this routine

   If maxValue < n Then

      fact(0) = 1

      For i = maxValue + 1 To n

         fact(i) = i * fact(i - 1)

      Next

      maxValue = n

   End If

   Factorial = fact(n)

End Function

Listing 1: Use a Smarter Factorial. This Factorial routine uses an internal table of results so it never needs to re-evaluate the same result. It also doesn’t waste time evaluating a result that the program never requests. Moreover, it is completely self-contained and can easily be reused in another application.
**

Use the And operator instead of Mod when the divisor is a number in the form 2^N. For instance, you can use two methods to extract the least significant byte in an Integer:

lowByte% = value% Mod 256

lowByte% = value% And 255

The second one is slightly faster.

**

You can often simplify If conditions that involve multiple Boolean operators. For instance, you can rewrite this code:

If x <> 0 Or y <> 0 Then ...

So it looks like this:

If x Or y Then ...

Similarly, you can simplify and optimize this test:

If x = 0 And y = 0 Then ...

So it looks like this:

If (x Or y) = 0 Then ...

Even the Xor Boolean operator can save you some overhead:

' this is the original expression

If (x = 0 And y = 0) Or _

   (x <> 0 And y <> 0) Then ...

' this is the optimized version

If (x = 0) Xor (y <> 0) Then ...
**

At times, you can replace an entire If...Else block with a simpler Boolean operation. For instance, you can replace this code:

If x > 0 Then y = 1 Else y = 0

with this faster, though more cryptic, code:

y = -(x > 0)

However, I suggest resorting to this kind of trick only if the replacement pays off in terms of overall execution speed, which might be the case only if the expression appears within a time-critical loop. Even in these circumstances, you should add a remark that clearly explains what your code is doing, so its future maintenance won’t become a nightmare for you or those who inherit this piece of code.

ways to speed Arrays

Reading and writing an item of an array is always slower than accessing a simple variable. Therefore, if you need to use the same array item in a loop repeatedly, you should assign the array item to a temporary variable and use that variable instead. I’ve included an example of this technique that scans an Integer array to check for duplicate values (see Listing 2). The routine is based on two nested loops and saves CPU time by caching the value of intArray(i) into a regular, nonarray variable. This simple trick makes the routine up to 80 percent faster. 

[Listing 2] VB3, VB4 16/32, VB5

Function AnyDuplicates(intArray() As Integer) As Boolean

   ' returns True if the array holds any duplicate values

   ' VB3 users: replace "As Boolean" with "As Integer"

   Dim i As Integer, j As Integer, lastItem As Integer

   Dim tmp As Integer

   ' evaluate UBound() only once

   lastItem = UBound(intArray)

   For i = LBound(intArray) To lastItem

      ' storing intArray(i) into a non-array variable

      ' saves an indexing operation within the inner 

      ' loop

      tmp = intArray(i)

      For j = i + 1 To lastItem

         If tmp = intArray(j) Then

            AnyDuplicates = True

            Exit Function

         End If

      Next

   Next

   AnyDuplicates = False

End Function

Listing 2: Find Duplicates Fast. This routine returns False only if an array does not include any duplicate values. When comparing each value with those that follow it, it stores the value into a regular, temporary variable in order to speed up the inner loop. Without this trick, the routine would be about 80 percent slower.
**

For the same reason, accessing items in a bidimensional matrix is slower than accessing itmes in a monodimensional array. Take this into account when designing your algorithms. For instance, when writing a computer checkers game, you might find it convenient to represent the board using an array of 64 items instead of a matrix of 8-by-8 cells.

**

If you must perform a lot of search operations in an array whose contents don’t change often, consider sorting the array and using an efficient binary search. On the other hand, sorting is a time-consuming technique, and you can often get better results using hash tables if you can afford the extra memory required. I won’t cover hash tables in this article, but you can use a hash table to optimize the AnyDuplicates routine that appears in Listing 2 (see Listing 3). As you see, the results can be astounding.

[Listing 3] VB3, VB4 16/32, VB5

Function AnyDuplicates2(intArray() As Integer) As Boolean

   ' returns True if the array holds any duplicate values

   ' VB3 users: replace "As Boolean" with "As Integer"

   Dim i As Long, value As Integer

   Dim numEls As Long, index As Long

   Const HASH_EMPTY = -32768

   ' use a hash table twice as large as the original 

   ' array - this will minimize the number of collisions

   numEls = (UBound(intArray) - LBound(intArray) + 1) * 2

   ReDim hashTable(0 To numEls - 1) As Integer

   ' init the hash table with the "empty" value

   ' note that the original array cannot include 

   ' the -32768 value

   For i = 0 To numEls - 1

      hashTable(i) = HASH_EMPTY

   Next

   For i = LBound(intArray) To UBound(intArray)

      ' read each element of the array and store it in 

      ' the hash table; the initial position is 

      ' evaluated using the simple expression:    

      '        index = value Mod numEls

      value = intArray(i)

      index = value Mod numEls

      ' scan the hash table for an empty slot

      Do Until hashTable(index) = HASH_EMPTY

         ' if we have found another item with the same

         ' value we can exit

         If hashTable(index) = value Then

            AnyDuplicates2 = True

            Exit Function

         End If

         ' test the next slot, but wrap around at the

         ' end of the array

         index = index + 1

         If index = numEls Then index = 0

      Loop

      ' we have found an empty slot, and can store

      ' the current value there

      hashTable(index) = value

   Next

   AnyDuplicates2 = False
End Function

Listing 3: Find Duplicates Faster. This routine solves the same problem as the routine in Listing 2, but uses a different approach based on hash tables. On a 1000-item array, this routine executes about 120 times faster than the original one; on a 10,000-item array, it executes more than 500 times faster. The only problem with this technique is that you have to reserve memory for a temporary array that is twice as large as the original one.
**

You can substitute an array of Integer values with a Byte array, provided the values to be stored in the array range from zero to 255. This technique doesn’t speed up your code directly, but it makes your program less resource-hungry, and therefore it performs better on older machines.

**

Copy a block of data among different arrays of the same type by using the CopyMemory API function instead of an explicit For...Next loop:

Declare Sub CopyMemory Lib "kernel32" _

   Alias "RtlMoveMemory" (dest _

   As Any, As Any, ByVal bytes _

   As Long)

' copy a() into b() - b() is of

' same type and properly DIMensioned

' N is the number of items to copy

CopyMemory b(0), a(0), n * Len(a(0)) 

You can use this method for arrays of any type, except for conventional variable-length strings, objects, and Type structures that embed string and object elements.

**

You can resort to the CopyMemory API function to shift a bunch of array items without a loop—for instance, after inserting or deleting a single item. I explained this technique in my article, “Get the Most Out of Your Arrays” [VBPJ July 1997].

**

When scanning all the items in a large bidimensional matrix, access them in a column-wise fashion. In other words, the outer For…Next loop should iterate on the columns, and the inner one should iterate on the rows. By doing so, you access array items that are at adjacent locations in memory (VB stores matrices one column after the other) and indirectly cause less paging activity. See my article, “Get the Most Out of Your Arrays” [VBPJ July 1997], for an example of a routine that becomes 10 times faster by simply inverting the order of the two nested For...Next loops.

**

Never use For Each on variant arrays. The usual For loop with an Integer or Long index is nearly twice as fast.

**

If you’re absolutely sure your VB5 program never raises a “Subscript out of range” error, you can compile it into native code and enable the “Remove Array Bounds Check” option. This could speed up your array-intensive routines by 50 percent or more.

untangle your String operations

Concatenation is a slow operation, and you should avoid it whenever you can. For instance, if you have to replace one or more characters in a string, use the Mid$ command:

Mid$(a$, 1, 1) = "A"

Don’t use this concatenation code:

a$ = "A" & Mid$(a$, 2)
**

Fixed-length strings are often slower than conventional variable-length strings. In fact, all string functions and operands can work only with variable-length strings. Therefore, all fixed-length strings first must be converted to temporary conventional strings before being processed. This might easily slow down your code by a factor of three or four.

**

Stay away from “$-less” string functions such as Left and Trim because they return a Variant value. Traditional functions such as Left$ and Trim$ return a string value you can use immediately within an expression without an implicit type conversion.

**

Dealing with Integer ANSI codes is always faster than manipulating characters. For instance, you can use these approaches to test whether the first character in a string is a space:

' comparison among character

If Left$(a$, 1) = " " Then ...

' comparison among numbers (40% faster)

If Asc(a$) = 32 Then ...

This approach is especially useful within Select Case blocks.

**

Use intrinsic string constants instead of Chr$() expressions. For instance, use vbTab instead of Chr$(9), and vbCrLf instead of Chr$(13) & Chr$(10).

**

Using the Len function to check whether a string contains characters is about 25 percent faster than an explicit comparison with a null string:

If Len(a$) = 0 Then ...
**

Many VB programmers compare strings in case-insensitive mode using LCase$ or UCase$ to convert both operands before the comparison. However, it’s faster to resort to the rarely used StrComp function:

If StrComp(a$, b$, vbTextCompare) _

   = 0 Then

      ' strings are equal

End If

**

Similarly, when you have to search a substring in a case-insensitive mode, use the InStr function with a fourth optional argument:

If Instr(1, a$, "vb", _

   vbTextCompare) Then ...

Note that in this syntax form ,you cannot omit the first argument—its omission prevents VB from compiling the statement. If your program performs only searches and comparisons in case-insensitive mode, you might add an “Option Compare Text” directive at the beginning of each module.

**

The InStr function lets you quickly test a single character against a list of characters. For example, with this code you can determine if a$ holds a vowel:

isVowel = InStr("aeiou", a$)
**

When you need to process each character in a string, you can assign the string to a Byte array and iterate on its elements. Remember, each Unicode character corresponds to two bytes. This approach is usually much faster because it saves a lot of Mid$ functions and thus a lot of temporary strings. Here’s a fast way to count the spaces in a string:

Dim b() as Byte, count As Integer

b() = source$

For i = 1 to UBound(b) Step LenB("A")

   If b(i) = 32 Then count = count + 1

Next

Note the unorthodox usage of the LenB() function. It returns 2 under VB4/32 and VB5, and 1 under VB4/16, so you can use the same code fragment without a #If compiler directive. Beware that this technique might not work if you localize your code for foreign countries that make use of the full Unicode character set.

**

The almost-forgotten Like operator can often save many CPU cycles by performing complex string tests in one operation:

' ID must be an alphabetical character

' followed by exactly three digits

If ID Like "[A-Z]###" Then ... 
faster Form Loading

If you can afford the memory overhead, hide a frequently used form instead of unloading it. The next time you show it, it will appear instantaneously. However, remember that in this case, the Form_Load event won’t fire, and you must manually initialize all the fields.

**

Don’t perform time-consuming operations, such as opening a database or filling a list box with many items, in the Form_Load event. It’s much better to wait for the Activate event. If your form is not modal, you must set a flag so you don’t re-initialize everything whenever the form loses and then gets the input focus:

Private Sub Form_Activate()

   Static initDone As Boolean

   If Not initDone Then

      ' open you DB here

      initDone = True

   End If

End Sub
**

You can speed up form loading—and reduce resource consumption at the same time—if you resort to fewer and simpler controls. For example, don’t use masked-edit controls if you can offer decent data-entry capabilities with regular text boxes. For the same reason, use a label with a border instead of a status bar or a read-only text box, a suite of Image controls instead of a Toolbar common control, a small scrollbar instead of a spin button, and so forth. 

**

Always set a form to Nothing after unloading it. This step frees all the memory allocated for its variables and arrays, and ensures that its Initialize event fires correctly the next time it’s referenced:

Unload Form1

Set Form1 = Nothing
Graphics tips give apps some zip

Setting AutoRedraw to False nearly doubles the speed of most graphic methods, and also reduces the memory footprint of the form.

**

If you need to set AutoRedraw to True, at least try to use fixed-border forms. In fact, when dealing with resizable forms, VB creates a persistent bitmap as large as the whole screen, even though the window is never maximized. Conversely, when dealing with a fixed-border form, VB allocates only enough memory for its current size.

**

If your form includes no graphic, or if it creates graphic shapes that don’t overlap with existing controls, set the ClipControls property to False to speed up the execution of all graphic methods.

**

Whether you embed images in forms at design time or use a resource file, you should use compressed bitmaps. Remember that the PictureBox control can read RLE bitmaps and WMF images, and its VB5 version also supports PCX and JPEG formats. If you used to resort to third-party controls to support these formats in your apps, you can get rid of them when you recompile the code under VB5, saving a significant amount of resources at run time.

use Controls intelligently

Moving or resizing a form or a control with the Move method is faster than modifying its individual Left, Top, Width, and Height properties.

**

If you have to reference a property of a control repeatedly—for instance, within a loop—you can assign the property to a temporary variable and use that variable in the loop:

tmp = Check1.Value

For i = 1 To Ubound(arr)

   arr(i) = tmp

Next 

This technique is sometimes referred to as property caching.

**

You can hide a control before setting many of its properties, and then make it visible again. This approach avoids a lot of redrawing activity and reduces flickering.

**

Under VB3, you can speed up your routines slightly by omitting the name of a property if it is the default property for that control. For example, use “Text1” instead of “Text1.Text”. This is not true under VB4 and VB5, both in p-code and in native-compiled code.

**

PictureBox controls consume a lot of resources. Unless you need their peculiar capabilities—such as support for DDE, graphic methods, and the ability to contain other objects—you can increase your app’s performance by replacing them with simpler Image controls.

**

Use metafile images instead of bitmap images within a PictureBox control, if applicable. Metafiles take fewer resources and usually redraw faster than bitmaps, unless you resize them.

don’t go with the flow: speed code Execution

Never use a Single, Double, or Variant variable as the controlling variable in For...Next loops. You can rewrite and optimize virtually any loop using an Integer or a Long as the controlling variable.

**

In VB5 applications compiled as native code, GoSubs are five to six times slower than calls to regular subs or functions. Conversely, they are considerably faster in p-code. This is one of the few examples of an optimization rule that doesn’t apply to both p-code and compiled applications.

**

You can split an If expression that includes several conditions tied with Boolean And operators into two or more If blocks. This technique often speeds execution because the second If is not executed if the first one fails. Here’s an example of a single If expression containing an And operator::

If x > 0 And Tan(x) < 1 Then y = 1

You can break down this code into separate If blocks:

If x > 0 Then

   If Tan(x) < 1 Then y = 1

End If

The revised code potentially reduces the number of calls to the Tan function, which is a relatively slow function. This technique is named short evaluation circuit. Most compilers—including the outdated VB-DOS compiler—automatically generate native code based on this optimization method. Unfortunately, the so-called VB5 “optimizing” compiler is not among them. I explained this technique in more detail in my article, “Soup Up Your 32-bit Apps” [VBPJ February 1996].

**

With Select Case and If...ElseIf blocks, always place first the Case clauses that deal with the most frequent values.

**

Don’t fill your code with unnecessary DoEvents statements, especially within time-critical loops. If you can’t avoid that, at least you can reduce the overhead by invoking DoEvents only every N iterations of the loop, using a statement like this:

If (loopNdx Mod 10) = 0 Then DoEvents

Alternatively, if you use DoEvents to trap mouse and keyboard activity, you can call it only if pending items are in the event queue. You can check this condition with the GetInputState API function:

Declare Function GetInputState Lib _

   "user32" Alias "GetInputState" () _

   As Long

   ' ...

   If GetInputState() Then DoEvents

**

Integer and Long arguments within a procedure are referenced slightly faster if you declare them in the argument list with ByVal. All other data types should be passed by reference. Never use ByVal when passing a string argument to a local procedure in your program.

**

Don’t use the IIf function in time-critical loops. A regular If...Then...Else block is always faster.

**

Don’t use an empty For...Next loop to add pauses to your programs. Instead, use the Sleep API function, which releases the CPU and lets other apps in the system effectively multitask:

Declare Sub Sleep Lib "kernel32" _

   (ByVal milliseconds As Long)

' pause for 5 seconds

Sleep 5000

Note this technique only works under VB4/32 and VB5; there is no equivalent function in the Win16 API.

**

Before the definitive compilation of an application, delete or comment out all the so-called dead code—routines and whole modules that your program never actually calls and that are often the residual of previous versions of the application. Spotting unused procedures and functions is not a trivial task, and for large projects, I suggest using a third-party cross-reference program that does it for you automatically. Or run the Code Profiler add-in and see which routines are never executed during a working session—be sure to test all of your app’s features.

reading from and writing to Files

VB4 and VB5 let you write an entire array to a disk file using a single Put statement:

Put #1, , strArray()

You can then use the Get statement to read that array back into memory:

ReDim strArray(numEls) As String

Get #1, , strArray()

Note that you must dimension the array properly before reading its contents from the file. This method works with any kind of array, including strings and user-defined types (UDTs)s, with the only exception of objects.

**

The fastest way to read a text file into a string variable or a text-box control is by using the Input$ function:

Text1.Text = Input$(#1, LOF(1))
**

When you write many small chunks of data to a file, it’s more efficient to gather all the items in a single string and write the string in a single operation:

' save the contents of an array

' of strings into a text file

For i = 1 To UBound(sArr)

   temp$ = temp$ & sArr(i) & vbCrLf

Next

Open "notes.txt" For Output As #1

Print #1, temp$

Close #1
Class notes on Objects and collections

You can always speed up the access to objects’ properties or methods by reducing the number of “dots” in the expression. For instance, when you reference the same “nested” object or control repeatedly:

Form1.Text1.Text = ""

Form1.Text1.ForeColor = 0

you can use a temporary object or a With clause:

With Form1.Text

   .Text = ""

   .ForeColor = 0

End With
**

Properties implemented as Public variables are always faster than a pair of Property Let/Get procedures, although the exact speed improvement depends on the language version. They are nearly eight times faster under VB4, but only about four times faster under VB5.

**

Within native-compiled VB5 apps, Friend properties and methods are up to six times faster than Public items, whether the class itself is Private or Public.

**

Referencing a collections’ item with a string key:

rootName = MyCollection("Root").Name

is faster than using a numerical index. Although this is true on average, numerical indexes are slightly faster when accessing the first items of the collection, but much slower when reading items located toward higher indexes.

**

For the same reason, always iterate on the items of a collection using a For Each...Next loop, which is up to 10 times faster than a regular For...Next loop with a numerical index.

**

When adding an item to a collection, don’t specify the before or after argument, unless you absolutely need to place the new element in a given position.

**

When you want to remove all items from a collection, don’t use a loop of Remove statements; simply set the collection to Nothing. This technique works only if no other object variables hold a reference to the same collection.

**

Auto-instancing objects, such as those created with Dim x As New Class1, are slightly slower than objects created by an explicit Set statement (Set x = New Class1). This is because each time VB references the object, it must also check whether the object needs to be instantiated first. This overhead causes a performance hit under VB4—up to 50 percent—but is almost negligible under VB5.

**

Stay away from generic Object variables and procedure parameters, which you can reference using late binding. Always use specific objects unless you are writing a procedure that has to work with different kind of objects.

OLE Automation

Generally speaking, calls to in-process DLL servers are much faster than out-of-process EXE servers, even by two orders of magnitude.

**

When passing arguments to in-process OLE servers, you can pass everything by reference because the server shares its client’s address space and can read the value of arguments directly. This technique is especially convenient when passing long strings. 

**

On the contrary, when passing arguments to out-of-process OLE servers, you should always pass everything by value, so OLE doesn’t resort to marshaling when the server returns the control to its client. Of course, this suggestion is void if you need to pass a ByRef argument for the OLE server to store a return value in it.

**

If you don’t need the additional capabilities offered by SingleUse OLE servers, you should stick to MultiUse servers, which take much less memory and fewer resources.

**

If your server has to inform its client apps that a given task is completed, use a callback mechanism rather than events implemented through the WithEvent clause. Events execute up to 30 times slower, don’t accept optional arguments or return values, and don’t work with remote servers.

Faster than a concorde: Jet and DAO

Use the simplest recordset type that supports all the features you need for a given task. For instance, if you only have to scan records to build a report, you might be satisfied with a forward-only, read-only recordset, which uses fewer resources and often works faster than a dynaset-type recordset.

**

With 500 records or fewer, snapshots are generally more efficient than dynasets, except when they include Memo and Large Binary fields. In the latter case, dynasets are more responsive because they download the data only when the VB code references the corresponding field.

**

Always filter your records using a suitable SELECT SQL statement rather than iterating on the whole recordset or table and skipping over the records you’re not interested in. Similarly, delete, insert, and modify records using action queries embedding DELETE, INSERT, and UPDATE SQL commands rather than by using DAO methods.

**

Compact your database often. This operation stores tables in adjacent pages of the database, rebuilds the indexes, updates the statistics on the database, and recompiles the queries.

**

Use parametrized QueryDef objects when you need to perform the same select or action query repeatedly. When DAO compiles a QueryDef object, it uses the most recent statistics to prepare its execution plan. Therefore, you should compact your database as often as you can to ensure the plan used for each query is based on the most recent data. 

**

Access data in remote databases by using attached tables rather than by opening them directly with an OpenDatabase method.

**

Always open your databases in read-only mode if you don’t need to update data. This practice prevents locking activity and speeds up both your program and all the other applications that access the same database concurrently.

**

For the best performance, wrap your updates within explicit transactions—a pair of BeginTrans and CommitTrans methods. However, be aware that pending transactions are stored in memory if possible, but too many pending transactions require the creation of a temporary database on disk, which causes a tremendous performance hit. Avoid this by performing a CommitTrans command and beginning another transaction every N records (see Listing 4). Because transactions enforce an implicit lock on data, never include user interface actions within a transaction, and always commit or roll back a transaction as soon as possible.

[Listing 4] VB4 [[16 and 32?]] VB5

' WS is the current workspace

' RS is the current recordset

ws.BeginTrans

Do Until rs.EOF

   recCount = recCount + 1

   If (recCount Mod 100) = 0 Then

      ' flush the transaction buffers every 100 records

      ws.CommitTrans

      ws.BeginTrans

   End If

   ' process the current record

   ' ...

   rs.MoveNext

Loop

' commit the last changes

ws.CommitTrans   

Listing 4: Explicit Transactions Speed Up Execution. DAO uses slow temporary database tables for a transaction with too many pending changes. Therefore, commit and restart a transaction many times during the loop, using the approach illustrated in this code fragment.
**

You can improve the performance of the Jet Engine 3.5 significantly by using the new SetOption command, which lets you override the default settings in the Windows Registry. For instance, you can control the size of the internal buffer (dbMaxBufferSize option), how often it’s flushed (dbSharedAsynchDelay and dbExclusiveAsynchDelay options), how often the engine attempts to put a lock on a page (dbLockDelay option), and how many attempts Jet performs (dbLockRetry option). For more information, see VB5’s online help.

**

If you’re not sure how long a query takes and how many records it returns, you can use a QueryDef object and set its MaxRecord property to a suitable value, such as 100 or 200 records. 

a direct flight: ODBCDirect and RDO

You can replicate on the client machine remote small tables that rarely change, which permits faster access. Your application should check whether a new version of the table is available on the server at the beginning of each execution and copy it locally if necessary.

**

You can use the QueryTimeout property of the rdoQuery object to avoid waiting too long for a query to finish. When the timeout expires, RDO fires a QueryTimeout event for the parent rdoConnection object, and you have a chance to cancel the query or continue to wait for its completion (see Listing 5).

[Listing 5] VB5

Option Explicit

Dim WithEvents cn As rdoConnection

Dim rs As rdoResultset

Private Sub Form_Load()

   ' show the form before opening the connection

   Show

   DoEvents

   ' open the connection

   Set cn = New rdoConnection

   With cn

      .Connect = "uid=;pwd=;DSN=WorkDB;"

      .CursorDriver = rdUseOdbc

      .EstablishConnection rdDriverNoPrompt

   End With

End Sub

Private Sub cmdStartQuery_Click()

   Dim sql As String

   sql = "Select * From Products Where price > 10"

   ' open the resultset running the query in asynch mode

   Set rs = cn.OpenResultset(sql, rdOpenKeyset, _

      rdConcurReadOnly, rdAsyncEnable)

End Sub

Private Sub cn_QueryComplete(ByVal Query As rdoQuery, _

   ByVal ErrorOccurred As Boolean)

      ' this event fires when the query is completed

      If ErrorOccurred Then

         MsgBox "An error is occurred while " _

            & "processing the query"

         ' in a real world program you should test            ' rdoErrors for a more elegant recovery

      Else

         MsgBox "The query has completed. " & _

            rs.RowCount & " records were returned."

      End If

End Sub

Private Sub cn_QueryTimeout(ByVal Query As rdoQuery, _

   Cancel As Boolean)

      If MsgBox("Query time-out. Do you wish to " & _

         "retry for additional " & cn.QueryTimeout & _

         " seconds?", vbYesNo) = vbYes Then

            ' the user is willing to wait

            Cancel = False

      End If

      ' no special action is required if the user 

      ' refuses to wait since the default value for 

      ' Cancel is True

End Sub

Listing 5: The Power of Events. You can use RDO events to build more responsive applications that run asynchronous queries and queries with a timeout that can be renewed if the user is willing to wait. Notice that the two features are independent of each other; you can run synchronous queries and still receive a QueryTimeout event that lets the operator decide whether the query should be terminated.
**

You can execute more responsive remote queries if you perform the so-called preconnection to an ODBC data source. When the program starts, open the remote database and close it immediately. This action doesn’t close the connection; the connection stays active for 10 minutes. You can change this default by acting on the ConnectTimeout value in the ...\Jet\3.5\Engines\ODBC key in the Registry. 

**

You can also keep connection usage as low as possible by working with dynasets of 100 records or fewer. In fact, dynasets with more than 100 records require two connections—one for the records, and one for the keys. However, other dynasets may share the latter connection for their own keys.

**

When using optimistic updates on a remote table with many fields, you can implement more efficient insert and delete operations by adding a timestamp field to the table, with a SQL command such as “ALTER TABLE RemoteTable ADD COLUMN VersionID TIMESTAMP.” Adding a timestamp field lets the database engine compare only the new VersionID field instead of the entire record to detect if the current record has been edited by another user. 

**

You can reduce network traffic significantly by implementing optimistic Client Batch cursors, now available in RDO 2.0. Moreover, you can release the connection temporarily, perform your locate updates to the recordset, reconnect again, and send the updates to the server. For more information, see VB5’s online help.

**

You can greatly improve the apparent speed of your client/server application by using asynchronous queries and connections (see Listing 5). 

CODE ONLINE

You can find all the code published in this issue of VBPJ on The Development Exchange (DevX) at http://www.windx.com. For details, please see “Get Extra Code in DevX’s Premier Club” in Letters to the Editor.

